A Shift in Time:

Using SAS® Date Alignment Operators

Andrew H. Karp

Sierra Information Services

Sonoma, California USA

www.**SierraInformation**.com

Copies of this Presentation

- Available for free download in PDF format from my web site
  - www.**SierraInformation**.com
  - Click on “Past Presentations” link



A Shift in Time

- Using SAS Date Alignment Operators
- Very powerful way to work with your date and time data.
- Re-align date and time intervals without tedious coding



A Shift in Time

- SAS supports the following **intervals**:
- **Date** variables
  - YEAR, SEMIYEAR, QTR, MONTH, SEMIMONTH
  - TENDAY, WEEK, WEEKDAY, DAY
- **Time** variables
  - HOUR, MINUTE, SECOND

Interval Name Construction

- **NAME*n.s***
  - **NAME**: basic **interval** type (see list on previous slide
  - ***N*** optional **multiplier** specifying that the interval is a *multiple of the basic interval type*
    - *YEAR4 consists of four year periods*
  - ***S*** optional **multiplier** shows the *starting subperiod index* specifying that the *intervals are shifted* to **later starting points.**
    - *YEAR4.3 = four year periods starting in March*

Examples of Intervals

Name	Result
YEAR	Year starting in January
YEAR.10	Year starting in October
YEAR4.11	Four year intervals starting in November of leap years (frequency of US presidential elections).
WEEK.2	Weeks starting on Monday
WEEK8.3	Eight week intervals starting on Tuesdays
HOURS8.8	Eight hour intervals starting at 7 am, 3 pm, and 11 pm

Example 1: Aggregate Work by Shift

```
 2 proc print data=expand.byhour2(where=(datepart(date) = '05jan1998'd));
 3 title 'A Shift in Time: Working with SAS Date and Time Intervals';
 4 title2 'Carpool Vehicles Crossing the Golden Gate Bridge by Hour on Jan 5, 1998';
 5 run;
 6
 7 proc expand data=expand.byhour2 from = hour to = hour8.8
 8      observed = total    out=by_shift;
 9 id date;
10 convert car_count=shift_count/transformin=(setmiss 0);
11 run;
12
13 options nodate nonumber nocenter;
14 proc print data=by_shift(where=(datepart(date) = '05jan1998'd));
15 title3 'Observations Aggregated by Shift';
16 title4 '2300-0700, 0700-1500, 1500-2300';
17 run;
```

Example 1: Aggregate Work by Shift

A Shift in Time: Working with SAS Date and Time Intervals
Carpool Vehicles Crossing the Golden Gate Bridge by Hour on Jan 5, 1998

Obs	DATE	car_count	date2	weekday
97	05JAN98:00:00:00	70	01/05/1998	2
98	05JAN98:01:00:00	64	01/05/1998	2
99	05JAN98:02:00:00	74	01/05/1998	2
100	05JAN98:03:00:00	61	01/05/1998	2
101	05JAN98:04:00:00	72	01/05/1998	2
102	05JAN98:05:00:00	74	01/05/1998	2
103	05JAN98:06:00:00	96	01/05/1998	2
104	05JAN98:07:00:00	83	01/05/1998	2
105	05JAN98:08:00:00	96	01/05/1998	2
106	05JAN98:09:00:00	74	01/05/1998	2
107	05JAN98:10:00:00	77	01/05/1998	2
108	05JAN98:11:00:00	73	01/05/1998	2
109	05JAN98:12:00:00	80	01/05/1998	2
110	05JAN98:13:00:00	72	01/05/1998	2
111	05JAN98:14:00:00	73	01/05/1998	2
112	05JAN98:15:00:00	61	01/05/1998	2
113	05JAN98:16:00:00	65	01/05/1998	2
114	05JAN98:17:00:00	70	01/05/1998	2
115	05JAN98:18:00:00	60	01/05/1998	2
116	05JAN98:19:00:00	64	01/05/1998	2
117	05JAN98:20:00:00	62	01/05/1998	2
118	05JAN98:21:00:00	63	01/05/1998	2
119	05JAN98:22:00:00	64	01/05/1998	2
120	05JAN98:23:00:00	69	01/05/1998	2

A Shift in Time: Working with SAS Date and Time Intervals
Carpool Vehicles Crossing the Golden Gate Bridge by Hour on Jan 5, 1998
Observations Aggregated by Shift
2300-0700, 0700-1500, 1500-2300

Obs	DATE	shift_count
14	05JAN98:07:00	628
15	05JAN98:15:00	509
16	05JAN98:23:00	554

Example2: Aggregate Hourly Observations to Weeks Starting on Mondays



```
proc expand data=sasclass.byhour2
     from = hour to = week.2
        observed = total
        out=expand.weekly;
id date;
convert car_count=weekly_count/
transformin=(setmiss 0);
run;
```

Example2: Aggregate Hourly Observations to Weekly, Starting on Mondays

Obs	DATE	Weekly Vehicle Crossings
250	Mon, 14 Oct 2002	7,654
251	Mon, 21 Oct 2002	7,895
252	Mon, 28 Oct 2002	7,902
253	Mon, 4 Nov 2002	8,253
254	Mon, 11 Nov 2002	8,545
255	Mon, 18 Nov 2002	8,513
256	Mon, 25 Nov 2002	8,933
257	Mon, 2 Dec 2002	9,282
258	Mon, 9 Dec 2002	9,441
259	Mon, 16 Dec 2002	9,527
260	Mon, 23 Dec 2002	9,547

Example 3:
Count the number of days, excluding Sundays, between 1/1/05 and 4/12/05



```
20 * example 3: count number of days, excluding Sundays, a store is open;
21 * between 1/1/05 and 4/12/05;
22 data _null_;
23 days_open = intck('weekday1w','01jan2005'd,'12apr2005'd);
24 put "Number of Days Open, Excluding Sunday, So Far in 2005:  " days_open;
25 run;
```

WEEKDAY1W means “Weekdays are Mondays through Saturdays, Weekend Days are Sundays”

Example 3:
Count the number of days, excluding Sundays, between 1/1/05 and 4/12/05

```
NOTE: SAS initialization used:
      real time               2.17 seconds
      cpu time                0.85 seconds

1     data _null_;
2     days_open = intck('weekday1w','01jan2005'd,'12apr2005'd);
3     put "Number of Days Open, Excluding Sunday, So Far in 2005:  " days_open;
4     run;

Number of Days Open, Excluding Sunday, So Far in 2005:  86
NOTE: DATA statement used (Total process time):
      real time               0.36 seconds
      cpu time                0.01 seconds
```



Example4: Counting US Federal Fiscal Years

```
data _null_;
fiscal_years =
intck('year.10','18dec1956'd,today());
put "Number Federal Fiscal Years Between";
put "My Birthday and &sysdate9 is:"
fiscal_years;
run;
```

Example 4: Counting Federal Fiscal Years

```
38    data _null_;
39    fiscal_years = intck('year.10','18dec1956'd,today());
40    put "Number Federal Fiscal Years Between";
41    put "My Birthday and &sysdate9 is:"  fiscal_years;
42    run;

Number Federal Fiscal Years Between
My Birthday and 21JUL2004 is:47
NOTE: DATA statement used (Total process time):
      real time           0.00 seconds
      cpu time            0.01 seconds
```



Example 5: Counting Two Week Periods Starting on Wednesdays

```
data _null_;
count = intck('week2.4','01jan2004'd,today());
put "Number of Two Week Periods Starting on Wednesday";
put "Between Jan 1, 2004 and &sysdate9 is: " count;
run;
```

Example 5: Counting Two Week Periods Starting on Wednesdays

```
43    data _null_;
44    count = intck('week2.4','01jan2004'd,today());
45    put "Number of Two Week Periods Starting on Wednesday";
46    put "Between Jan 1, 2004 and &sysdate9 is: " count;
47    run;

Number of Two Week Periods Starting on Wednesday
Between Jan 1, 2004 and 21JUL2004 is: 14
NOTE: DATA statement used (Total proces    time):
      real time                0.00 seconds
      cpu time                 0.00 seconds
```



Example 6: Using Date Alignment Operators with PROCs GPLOT and PLOT

- The HAXIS Option in PROCs GPLOT and PLOT specifies both the horizontal range and tick marks.
  - Date Constants and a Date Interval Can Be Given in the Option

Example 6: Using Date Alignment Operators with PROCs GPLOT and PLOT



Example 6: Using Date Alignment Operators with PROCs GPLOT and PLOT



```
options orientation=landscape;
symbol1 color=blue line=1 value=circle interpol=join;
* example 6: Use Alignment Operators in the HAXIS Option;
proc gplot data=series1;
plot events*month/haxis = '01jan1992'd to '01dec2004'd by year;
title "A Shift in Time: Working with SAS Date Alignment Operators";
Title2 "Using Date Alignment Operators in the HAXIS Option";
title3 "haxis = '01jan1992'd to '01dec2004'd by year";
run;
quit;
```

Example 6: Using Date Alignment Operators with PROCs GPLOT and PLOT



Example 6: Using Date Alignment Operators with PROCs GPLOT and PLOT

```
35 options orientation=landscape;
36 symbol1 color=blue line=1 value=circle interpol=join;
37 * example 6: Use Alignment Operators in the HAXIS Option;
38 proc gplot data=series1(where=(month GE '01jun1992'd));
39 plot events*month/haxis = '01jun1992'd to '01jun2005'd by year.6;
40 title "A Shift in Time: Working with SAS Date Alignment Operators";
41 Title2 "Using Date Alignment Operators in the HAXIS Option";
42 title3 "haxis = '01jun1992'd to '01dec2004'd by year.6";
43 title4 "Fiscal Years Starting in June";
44 run;
45 quit;
```

Example 6: Using Date Alignment Operators with PROCs GPLOT and PLOT



Example 6: Using Date Alignment Operators with PROCs GPLOT and PLOT

```
35 options orientation=landscape;
36 symbol1 color=blue line=1 value=circle interpol=join;
37 * example 6: Use Alignment Operators in the HAXIS Option;
38 proc gplot data=series1(where=(month GE '01jun1992'd));
39 plot events*month/haxis = '01jun1992'd to '01jun2005'd by year2.6;
40 title "A Shift in Time: Working with SAS Date Alignment Operators";
41 Title2 "Using Date Alignment Operators in the HAXIS Option";
42 title3 "haxis = '01jun1992'd to '01dec2004'd by year2.6";
43 title4 "Two-Year Periods Starting in June";
44 run;
45 quit;
```

Example 6: Using Date Alignment Operators with PROCs GPLOT and PLOT



Summary and Conclusions

- SAS Date and Time Alignment Operators
  - Very Powerful
  - Easy to Implement
  - Avoid Need for Tedious Data Step Coding
  - Enhance Value of Reports/Analyses

Thank You Attending SUGI 30 and the Coder's Corner



- Questions?
- Comments?
- Copies of this presentation available at
- www.**SierraInformation**.com

Disclaimer

- The information in this document is presented “as is” without any warranty of merchantability, usability, or fitness for a particular purpose.
- Sierra Information Services is not able to provide technical support or advice on the topic(s) presented in this presentation.



Copyright Information